QT += widgets uiplugin
CONFIG += plugin
CONFIG += c++14
TEMPLATE = lib
DEFINES += FILTERPLUGINDESIGNER_LIBRARY

TARGET = $$qtLibraryTarget($$TARGET)
INSTALLS += target

windows {
    target.path = $$(QTDIR)/../../Tools/QtCreator/bin/plugins/designer
    debug:target_lib.files = $$OUT_PWD/debug/$${TARGET}.lib
    release:target_lib.files = $$OUT_PWD/release/$${TARGET}.lib
    target_lib.path = $$(QTDIR)/../../Tools/QtCreator/bin/plugins/designer
    INSTALLS += target_lib

    INCLUDEPATH += $$(OPENCV_HOME)/include
    LIBS += -L$$(OPENCV_HOME)/x86/vc14/lib -lopencv_core2413 -lopencv_imgproc2413
}

HEADERS += \
    Filter.h \
    FilterBlur.h \
    FilterGrayscale.h \
    FilterOriginal.h \
    FilterPluginDesigner.h \
    FilterWidget.h \
    filter_plugin_designer_global.h

SOURCES += \
    Filter.cpp \
    FilterBlur.cpp \
    FilterGrayscale.cpp \
    FilterOriginal.cpp \
    FilterPluginDesigner.cpp \
    FilterWidget.cpp

FORMS += \
    FilterWidget.ui

RESOURCES += \
    filter_plugin_designer.qrc
